

# Analytical Solutions of PDEs using PDEtools in Maple

Aleksandar Donev, Courant Institute

This is largely based on examples in the excellent Maple documentation

**> restart:**

The PDEtools package is a collection of commands and routines for finding analytical solutions for [partial differential equations](#) (PDEs) based on the paper "A Computational Approach for the Analytical Solving of Partial Differential Equations" by E.S. Cheb-Terrab and K. von Bulow (see [References](#)), and the continuation of this work by the same authors ([Symmetry routines](#)) during 2004.

**> with(PDEtools):**

## First-Order PDEs

Maple knows about the method of characteristics:

**> U := diff\_table(u(x,t)): # To enable compact notation U\_xx etc.**

**> PDE := U[t]+c\*U[x]=-lambda\*U[]; # Advection-reaction linear equation**

$$PDE := c \left( \frac{\partial}{\partial x} u(x, t) \right) + \frac{\partial}{\partial t} u(x, t) = -\lambda u(x, t) \quad (1)$$

**> IC := eval(U[], t=0) = phi(x); # Initial conditions**

$$IC := u(x, 0) = \phi(x) \quad (2)$$

**> pdsolve([PDE,IC]); # Try to solve the PDE automatically - works!**

$$u(x, t) = \phi(-ct + x) e^{-\lambda t} \quad (3)$$

Try to get a general solution:

**> pdsolve(PDE); # No initial condition**

$$u(x, t) = \_FI \left( \frac{ct - x}{c} \right) e^{-\frac{\lambda x}{c}} \quad (4)$$

**> PDE := U[t]+c\*U[x]=-lambda\*U[]^2; # Advection-reaction nonlinear equation**

$$PDE := c \left( \frac{\partial}{\partial x} u(x, t) \right) + \frac{\partial}{\partial t} u(x, t) = -\lambda u(x, t)^2 \quad (5)$$

**> pdsolve([PDE,IC]); # Works here also since simple first-order equation**

$$u(x, t) = \frac{\phi(-ct + x)}{\phi(-ct + x) \lambda t + 1} \quad (6)$$

**> PDE := U[t]+c\*U[]\*U[x]=0; # Burger's equation**

$$PDE := c u(x, t) \left( \frac{\partial}{\partial x} u(x, t) \right) + \frac{\partial}{\partial t} u(x, t) = 0 \quad (7)$$

**> pdsolve([PDE,IC]); # Does not return anything (cannot find solution)**

## Heat Equation

**> restart: with(PDEtools):**

Maple knows about the method of separation of variables:

**> BVP:=[diff(u(x, t), t) = k\*diff(u(x, t), x, x), u(0, t) = 0, u**

`(Pi, t) = 0, u(x,0) = f(x)]; # Heat equation in 1D with homogeneous Dirichlet BCs`

$$BVP := \left[ \frac{\partial}{\partial t} u(x, t) = k \left( \frac{\partial^2}{\partial x^2} u(x, t) \right), u(0, t) = 0, u(\pi, t) = 0, u(x, 0) = f(x) \right] \quad (8)$$


`> sol:=pdsolve(BVP);`

$$sol := u(x, t) = \sum_{n=1}^{\infty} \frac{2 \left( \int_0^{\pi} f(x) \sin(nx) dx \right) \sin(nx) e^{-k n^2 t}}{\pi} \quad (9)$$

`> g(x):=piecewise(x<1/2,0,x>Pi/2,Pi,x=Pi/2,Pi/2);`

$$g(x) := \begin{cases} 0 & x < \frac{1}{2} \\ \pi & \frac{1}{2} \pi < x \\ \frac{1}{2} \pi & x = \frac{1}{2} \pi \end{cases} \quad (10)$$

`> plot(g(x),x=0..Pi)`



`> pdsolve(eval(BVP, f(x)=g(x)));`

Error. (in casesplit/K) this version of casesplit is not yet handling the function: piecewise

> temp:=eval(sol, f(x)=g(x));

$$temp := u(x, t) = \sum_{Z1=1}^{\infty} \frac{2 \int_0^{\pi} \begin{cases} 0 & x < \frac{1}{2} \\ \pi & \frac{1}{2} \pi < x \\ \frac{1}{2} \pi & x = \frac{1}{2} \pi \end{cases} \sin(Z1 x) dx}{\pi} \sin(Z1 x) e^{-k_{Z1}^2 t} \quad (11)$$

> sol\_series:=value(temp);

$$sol\_series := u(x, t) = \sum_{Z1=1}^{\infty} \frac{2 \left( \cos\left(\frac{1}{2} \pi_{Z1}\right) - (-1)^{-Z1} \right) \sin(Z1 x) e^{-k_{Z1}^2 t}}{Z1} \quad (12)$$

## Numerical Evaluation

### Plotting the Solution

To plot this we need to truncate the sum to a few terms

> summand:=op(1, rhs(sol\_series));

$$summand := \frac{2 \left( \cos\left(\frac{1}{2} \pi_{Z1}\right) - (-1)^{-Z1} \right) \sin(Z1 x) e^{-k_{Z1}^2 t}}{Z1} \quad (13)$$

> summand:=eval(summand, {\_Z1=n, k=1});


$$summand := \frac{2 \left( \cos\left(\frac{1}{2} \pi n\right) - (-1)^n \right) \sin(n x) e^{-n^2 t}}{n} \quad (14)$$

Here we only take 10 terms:

> sol\_approx:=unapply(simplify(add(summand, n=1..10)),t);

$$sol\_approx := t \rightarrow 2 \sin(x) e^{-t} - 2 \sin(2 x) e^{-4t} + \frac{2}{3} \sin(3 x) e^{-9t} + \frac{2}{5} \sin(5 x) e^{-25t} \\ - \frac{2}{3} \sin(6 x) e^{-36t} + \frac{2}{7} \sin(7 x) e^{-49t} + \frac{2}{9} \sin(9 x) e^{-81t} - \frac{2}{5} \sin(10 x) e^{-100t} \quad (15)$$


> plot( {seq(sol\_approx(i\*0.05), i=0..5)}, x=0..Pi);



We can turn this into an animation to make it more informative:

```
> plots[animate]( plot, [sol_approx(t), x=0..Pi], t=0..2 );
```

$t = 0.75000$



### ***Gibbs Phenomenon***


At  $t=0$  we are getting the Gibbs phenomenon, lets explore this some more:

```
> summ0:=eval(summand, t=0);
```

$$summ0 := \frac{2 \left( \cos\left(\frac{1}{2} \pi n\right) - (-1)^n \right) \sin(n x)}{n}$$

(16)

```
> plot( {seq(add(summ0, n=1..4*N), N=1..5)}, x=0..Pi);
```




### ***Numerical Accuracy***

To see how good our solution is, we should really compare with the solution obtained by including more terms

```
> sol_better_approx:=unapply(simplify(add(summand, n=1..20)),t):
> plots[animate]( plot, [{sol_approx(t), sol_better_approx(t)}, x=
0..Pi], t=0..0.1, frames=50);
```

$t = 0.0081633$



## Wave Equation (from recitation)

Here we try to solve Problem 9 in section 4.7 of APDE, as was done in recitation on 4/15/2016 by recitation leader JT. Look at his solution notes first.

> restart: with(PDEtools):

> BVP := [diff(u(x, t), t, t) = diff(u(x, t), x, x), u(0, t) = 0, u(1, t) = sin(t), u(x, 0) = x\*(1-x), D[2](u)(x, 0) = 0];

$$BVP := \left[ \frac{\partial^2}{\partial t^2} u(x, t) = \frac{\partial^2}{\partial x^2} u(x, t), u(0, t) = 0, u(1, t) = \sin(t), u(x, 0) = x(1-x), D_2(u)(x, 0) = 0 \right] \quad (17)$$

> #infolevel[pdsolve]:=5; # If you want to see what Maple is trying out

> sol := pdsolve(BVP); # No solution returned

sol :=

(18)

Let's try to see what the problem is, inhomogeneous BCs, or initial condition, or what:

> BVP\_hom := [diff(u(x, t), t, t) = diff(u(x, t), x, x), u(0, t) = 0, u(1, t) = 0, u(x, 0) = x\*(1-x), D[2](u)(x, 0) = 0]; # Make the BC at x=1 be homogeneous

$$BVP\_hom := \left[ \frac{\partial^2}{\partial t^2} u(x, t) = \frac{\partial^2}{\partial x^2} u(x, t), u(0, t) = 0, u(1, t) = 0, u(x, 0) = x(1-x), D_2(u)(x, 0) = 0 \right] \quad (19)$$

> sol:=pdsolve(BVP\_hom); # Now it worked fine

$$sol := u(x, t) = \sum_{Z1=1}^{\infty} \left( -\frac{4((-1)^{Z1} - 1) \sin(Z1 \pi x) \cos(Z1 \pi t)}{\pi^3 Z1^3} \right) \quad (20)$$

So let's break up the problem into two problems just like done in class:

First a simple Laplace equation to take care of the BC. Note that here we tell Maple the solution is only a function of x, since for each t we are solving only an equation for x. If we wrote v(x,t) below it wouldn't work even though the same solution would work:

> BVP\_BC:= [0 = diff(v(x), x, x), v(0) = 0, v(1) = sin(t)];

$$BVP\_BC := \left[ 0 = \frac{d^2}{dx^2} v(x), v(0) = 0, v(1) = \sin(t) \right] \quad (21)$$

> sol\_BC:=pdsolve(BVP\_BC);

$$sol\_BC := v(x) = \sin(t) x \quad (22)$$

> v(x,t):=rhs(sol\_BC);

$$v(x, t) := \sin(t) x \quad (23)$$

Now solve with homogeneous BCs but make sure to subtract the time derivative of the v(x,t) from the right hand side as a source term:

> BVP\_IC:= [diff(u(x, t), t, t) = diff(u(x, t), x, x) - diff(v(x, t), t), u(0, t) = 0, u(1, t) = 0, u(x, 0) = x\*(1-x) - eval(v(x, t), t=0), D[2](u)(x, 0) = 0];

$$BVP\_IC := \left[ \frac{\partial^2}{\partial t^2} u(x, t) = \frac{\partial^2}{\partial x^2} u(x, t) - \cos(t) x, u(0, t) = 0, u(1, t) = 0, u(x, 0) = x(1-x), D_2(u)(x, 0) = 0 \right] \quad (24)$$

> pdsolve(BVP\_IC); # Sadly, it again does not return an answer!

Seems Maple hasn't learned how to solve this sort of problem with forcing yet, at least not automatically. We can still use Maple to help us with some of the integrals, however:

> a\_k:=simplify(2\*int(x\*sin(k\*Pi\*x),x=0..1)) assuming k, integer; # Eq. (9) in JT's solution

$$a_k := \frac{2(-1)^{1+k}}{k\pi} \quad (25)$$

> b\_k:=simplify(2\*int(x\*(1-x)\*sin(k\*Pi\*x),x=0..1)) assuming k, integer; # Eq. (10) in JT's solution

$$b_k := \frac{4(-1)^{1+k} + 4}{k^3 \pi^3} \quad (26)$$

> sol\_ODE:=dsolve({diff(T(t),t,t)+(k\*Pi)^2\*T(t)=a\_k\*sin(t),T(0)=b\_k,D(T)(0)=-a\_k}, T(t)); # Eqs. (14,15,23) in JT's solution, the most tedious step

(27)



$$\begin{aligned} \text{sol\_ODE} := T(t) = & -\frac{2 \sin(k \pi t) (-1)^{1+k}}{\pi^2 k^2 - 1} + \frac{4 \cos(k \pi t) ((-1)^{1+k} + 1)}{k^3 \pi^3} \\ & + \frac{2 (-1)^{1+k} \sin(t)}{k \pi (\pi^2 k^2 - 1)} \end{aligned} \quad (27)$$

## Giving Hints

For complicated nonlinear PDEs usually Maple won't know what to do (perhaps no one does!)

> **V := diff\_table(v(x,y)):**

> **PDE := V[\*]V[x,y] + V[x]\*V[y] = 1;**

$$PDE := v(x, y) \left( \frac{\partial^2}{\partial y \partial x} v(x, y) \right) + \left( \frac{\partial}{\partial x} v(x, y) \right) \left( \frac{\partial}{\partial y} v(x, y) \right) = 1 \quad (28)$$

> **pdsolve(PDE); # Maple tried separation of variables and it worked!**

$$(v(x, y) = \_F1(x) \_F2(y)) \&\text{where} \left[ \left[ \frac{d}{dx} \_F1(x) = \frac{-c_1}{\_F1(x)}, \frac{d}{dy} \_F2(y) = \frac{1}{2 \_F2(y) \_c1} \right] \right] \quad (29)$$

> **pdsolve(PDE, HINT=X(x)\*Y(y)); # Give hint to use separation of variables**

$$(v(x, y) = X(x) Y(y)) \&\text{where} \left[ \left[ \frac{d}{dx} X(x) = \frac{-c_1}{X(x)}, \frac{d}{dy} Y(y) = \frac{1}{2 Y(y) \_c1} \right] \right] \quad (30)$$

Maple did not by default try to solve the ODEs above, but we can ask it to continue solving:

> **pdsolve(PDE, HINT=X(x)\*Y(y), build);**

$$v(x, y) = \frac{\sqrt{2 x \_c1 + \_C1} \sqrt{\_C2 \_c1^2 + y \_c1}}{\_c1} \quad (31)$$

> **pdsolve(PDE, HINT=sqrt(P(x,y))); # Find an even more general solution**

$$v(x, y) = \sqrt{\_F2(x) + \_F1(y) + 2 x y} \quad (32)$$

## Solitons in the Korteweg–de Vries Equation

Now we try an equation we have not seen in class, one that exhibits solutions called solitons, which are stable traveling waves that find uses in fiber optic communication, and can be observed in certain water channels.

> **restart: with(PDEtools):**

> **U := diff\_table(u(x,t)): # To enable compact notation U\_xx etc.**

The KdV equation is a nonlinear third-order PDE that adds a third derivative term to the Burgers equation

> **PDE := U[t]+6\*U[\*]U[x]+U[x,x,x]=0; # The KdV equation with coefficient 6 chosen for convenience**

$$PDE := 6 u(x, t) \left( \frac{\partial}{\partial x} u(x, t) \right) + \frac{\partial}{\partial t} u(x, t) + \frac{\partial^3}{\partial x^3} u(x, t) = 0 \quad (33)$$

> **infolevel[pdsolve]:=3:**

> **sol:=pdsolve(PDE); # Looks like Maple knew to look for traveling wave solutions**

First set of solution methods (general or quasi general solution)  
 Second set of solution methods (complete solutions)  
 Third set of solution methods (simple HINTs for separating variables)  
 PDE linear in highest derivatives - trying a separation of variables by \*  
 HINT = \*  
 Fourth set of solution methods  
 Preparing a solution HINT ...  
 Trying HINT = \_F1(x)\*\_F2(t)  
 Fourth set of solution methods  
 Preparing a solution HINT ...  
 Trying HINT = \_F1(x)+\_F2(t)  
 Trying travelling wave solutions as power series in tanh ...  
 \* Using tau = tanh(t\*C[2]+x\*C[1]+C[0])  
 \* Equivalent ODE system: {-C[1]^3\*(tau^2-1)\*(tau^4-2\*tau^2+1)\*diff(diff(diff(u(tau),tau),tau),tau)-C[1]^3\*(tau^2-1)\*(6\*tau^3-6\*tau)\*diff(diff(u(tau),tau),tau)+(-6\*u(tau)\*C[1]\*(tau^2-1)-C[2]\*(tau^2-1)-C[1]^3\*(tau^2-1)\*(6\*tau^2-2))\*diff(u(tau),tau)}  
 \* Ordering for functions: [u(tau)]  
 \* Cases for the upper bounds: [[n[1] = 2]]  
 \* Power series solution [1]: {u(tau) = tau^2\*A[1,2]+tau\*A[1,1]+A[1,0]}  
 \* Solution [1] for {A[i, j], C[k]}: [[A[1,1] = 0, A[1,2] = 0], [A[1,0] = 1/6\*(8\*C[1]^3-C[2])/C[1], A[1,1] = 0, A[1,2] = -2\*C[1]^2]]  
 travelling wave solutions successful.

$$sol := u(x, t) = -2\_C2^2 \tanh(\_C2 x + \_C3 t + \_C1)^2 + \frac{1}{6} \frac{8\_C2^3 - \_C3}{\_C2} \quad (34)$$

> soliton:=u(x,t)=1/2\*c\*(sech(sqrt(c)/2\*(x-c\*t)))^2;

$$soliton := u(x, t) = \frac{1}{2} c \operatorname{sech}\left(\frac{1}{2} \sqrt{c} (-ct+x)\right)^2 \quad (35)$$

> convert(soliton, expln);

$$u(x, t) = \frac{2c}{\left(e^{\frac{1}{2}\sqrt{c}(-ct+x)} + e^{-\frac{1}{2}\sqrt{c}(-ct+x)}\right)^2} \quad (36)$$


> pdetest(soliton,PDE); # Check if it a solution of the PDE

0 (37)

> sol\_example:=eval(rhs(soliton),c=1);

$$sol\_example := \frac{1}{2} \operatorname{sech}\left(\frac{1}{2} t - \frac{1}{2} x\right)^2 \quad (38)$$

> plots[animate](plot, [{sol\_example}, x=-10..50], t=0..50, frames=50);



**> ODE:=eval(eval(PDE, u(x,t)=f(x-c\*t)), x-c\*t=xi);**  

$$ODE := 6f(\xi) D(f)(\xi) - D(f)(\xi) c + D^{(3)}(f)(\xi) = 0 \quad (39)$$

**> ODE:=convert(ODE,diff);**  

$$ODE := 6f(\xi) \left( \frac{d}{d\xi} f(\xi) \right) - \left( \frac{d}{d\xi} f(\xi) \right) c + \frac{d^3}{d\xi^3} f(\xi) = 0 \quad (40)$$

**> simpler\_ODE:=int(lhs(ODE),xi)-A=0;**  

$$simpler\_ODE := 3f(\xi)^2 - cf(\xi) + \frac{d^2}{d\xi^2} f(\xi) - A = 0 \quad (41)$$

**> sols:=dsolve(simpler\_ODE,f(xi)); # There are many solutions and they are not simple**

$$sols := \int \frac{1}{\sqrt{-2_a^3 + _a^2 c + 2A_a + _C1}} d_a - \xi - _C2 = 0, \int \left( \frac{1}{\sqrt{-2_a^3 + _a^2 c + 2A_a + _C1}} \right) d_a - \xi - _C2 = 0 \quad (42)$$

Let's give up on analytical solutions and just use numerics to explore our special soliton solutions...next

Lclass